Fachhochschule Nordwestschweiz
Hochschule für Wirtschaft

Pi

Berechnung der ersten Tausend Stellen der Kreiszahl π auf einem PC mit Excel Visual Basic

Studie

Autor: Helmut Vetter
Ort, Datum: Arlesheim, 31.08.2014

Diese Arbeit wurde mit TexLive erstellt.

Pi
Berechnung der ersten Tausend Stellen der Kreiszahl π auf einem PC mit Excel Visual Basic

Autor
Vetter, Helmut
Schillerweg 2
CH-4144 Arlesheim
061 599 51 09
helmut.vetter@fhnw.ch

Auftraggeberschaft
Fachhochschule für Wirtschaft
Tanner, Christian

Arlesheim, August 2014

Ehrenwörtliche Erklärung

Ich versichere, dass ich die vorliegende Arbeit selbstständig und ohne Benutzung anderer als der im Literaturverzeichnis angegebenen Quellen und Hilfsmittel angefertigt habe.

Die wörtlich oder inhaltlich den im Literaturverzeichnis aufgeführten Quellen und Hilfsmitteln entnommenen Stellen sind in der Arbeit als Zitat bzw. Paraphrase kenntlich gemacht.

Diese Arbeit ist noch nicht veröffentlicht worden. Sie ist somit weder anderen Interessenten zugänglich gemacht noch einer anderen Prüfungsbehörde vorgelegt worden.

Arlesheim, 31.08.2014

Helmut Vetter

Helmut Vetter

Management Summary

Jeder Mensch lernt in der Schule, dass Pi das Verhältnis von Kreisumfang u zu Kreisdurchmesser d ist: $\pi = p := \frac{u}{d}$. Ebenso kennt jeder den Näherungswert $\pi \approx 3.14$.

Nimmt man eine Dose mit kreisrundem Boden (Migros Erbsen fein), so kann man durch Abrollen auf einem Tisch deren Umfang recht genau messen: $u = 23.0$cm. Auch der Durchmesser kann recht genau abgemessen werden: $d = 7.3$cm. Die Genauigkeit bei beiden Messungen dürfte im Bereich von $\pm$1mm liegen.
Man schreibt $u = 23.0\text{cm} \pm 0.1\text{cm}$ bzw. $d = 7.3\text{cm} \pm 0.1\text{cm}$.
Die Fehlerrechnung lehrt, dass der mittlere relative Fehler des Resultats $\frac{\Delta p}{p}$ gleich $\sqrt{(\frac{\Delta u}{u})^2 + (\frac{\Delta d}{d})^2} \geq \max(\frac{\Delta u}{u}, \frac{\Delta d}{d})$ ist.
In unserm Beispiel ist $\frac{\Delta u}{u} = \frac{0.1\text{cm}}{23.0\text{cm}} = 4.3\,{}^0\!/_{00}$ und $\frac{\Delta u}{u} = \frac{0.1\text{cm}}{7.3\text{cm}} = 13.7\,{}^0\!/_{00}$, also $\frac{\Delta p}{p} \geq 13.7\,{}^0\!/_{00}$.
Bei einem Wert $p = \frac{u}{d} = \frac{23.0}{7.3} = 3.1507$ ergibt sich eine Ungenauigkeit (=mittlerer absoluter Fehler) von $\Delta p \geq 13.7\,{}^0\!/_{00} \cdot 3.1507 = 0.0432$, was also schon die zweite Hinterkommastelle zu einer komplett unsicheren Sache macht!

Dieses Problem zeigt sich bei jeder realen Messung:
Auch eine Messung auf Atome genau liefert nur eine sehr begrenzte Stellenzahl von π: Der Durchmesser eines Atomes beträgt etwa 1 Angström $= 10^{-10}$m. Der Relative Fehler von d ist also $\frac{10^{-10}\text{m}}{0.073\text{m}} \geq 1.37 \cdot 10^{-9}$, was sich in einem relativen Fehler $\frac{\Delta p}{p} \geq 1.37 \cdot 10^{-9}$ und also $\Delta p \geq 4.3 \cdot 10^{-9}$ wiederspiegelt. Das heisst die Zahl an der 9-ten Hinterkommastelle des so errechneten π ist bereits unsicher.

Im Umkehrschluss bedeutet das, dass für reale Rechnungen eine Kenntnis von π auf 10 Stellen völlig ausreichend ist. Excel verwendet $\pi \approx 3.14159265358979$, eine Rundung auf 14 Hinterkommastellen.

Mit analytischen Hilfsmitteln kann man zeigen, dass π keine rationale Zahl ist; das heisst Pi ist nicht als Bruch zweier ganzer Zahlen darstellbar. Die Konsequenz daraus ist, dass π keine abbrechende und auch keine periodische Dezimalbruchdarstellung hat: 3.141592653589793238462643383279502884197169399...

In diesem Paper wird mit Hilfsmitteln aus der Analysis, einer Programmierumgebung (Visual Basic) und einem Input-Output-Terminal (Excel) π auf Tausend Stellen berechnet.

Sie erhalten dabei Einblicke in die Wirkungsweise der Analysis und werden sehen wie elegant und zuverlässig Programmcodes es ermöglichen, aufwändige Rechnungen durchzuführen.

Inhaltsverzeichnis

1 Einleitung

1 Ziel ist es die ersten 1000 Stellen der Kreiszahl π auf einem Personalcomputer zu berechnen.

2 Theoretische Überlegungen

2 Die Ableitung der Funktion $f :]\frac{-\pi}{2}, \frac{\pi}{2}[\rightarrow \mathbb{R}$ definiert durch $f(x) := \tan(x) = \frac{\sin(x)}{\cos(x)}$ ist

$$f'(x) = \frac{\cos(x) \cdot \cos(x) - \sin(x) \cdot (-\sin(x))}{\cos^2(x)} = \frac{\cos^2(x) + \sin^2(x)}{\cos^2(x)} = 1 + \tan^2(x)$$

3 Die Ableitung der Umkehrfunktion $f^{-1}(x) = \arctan(x)$ ist

$$(f^{-1}(x))' = \frac{1}{f'(f^{-1}(x)} = \frac{1}{1 + \tan^2(\arctan(x))} = \frac{1}{1 + x^2}$$

4 Daraus ergibt sich die für unsere Berechnung fundamentale Beziehung zwischen einer rationalen und einer trigonometrischen Funktion (unter Verwendung von $\arctan(0) = 0$):

$$\int_0^x \frac{1}{1 + t^2} \cdot dt = \arctan(x) \text{ für alle } x \in \mathbb{R}$$

5 Der Integrand $\frac{1}{1 + t^2}$ lässt sich für $|t| < 1$ als geometrische Reihe darstellen:

$$\frac{1}{1 + t^2} = 1 - t^2 + t^4 - t^6 + \ldots = \sum_{k=0}^{\infty} (-1)^k \cdot t^{2k}$$

6 Wegen der gleichmässigen Konvergenz der Reihe auf jedem Intervall $[-Q, Q]$ für $0 < Q < 1$ darf die Reihe gliedweise intergriert werden mit dem Resultat:

$$\arctan(x) = x - \frac{x^3}{3} + \frac{x^5}{5} - \frac{x^7}{7} + \ldots = \sum_{k=0}^{\infty} (-1)^k \cdot \frac{x^{2k+1}}{2k + 1} \text{ für } |x| < 1$$

7 Nach dem Satz von Abel stimmt die Gleichung sogar im Randpunkt $x = 1$.

8 Von geometrischer Seite her gilt:

$$\arctan(1) = \frac{\pi}{4}$$

9 Somit ergibt sich die bekannte Identität von Leibniz:

$$1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} + \ldots = \frac{\pi}{4}$$

10 Diese ist wegen der langsamen Konvergenz ungeeignet um viele Stellen von π zu berechnen.

11 Für den Tangens einer Summe von zwei Winkeln ergibt sich die folgende Rückführung auf die Tangens der einzelnen Winkel:

$$\tan(\alpha + \beta) = \frac{\sin(\alpha + \beta)}{\cos(\alpha + \beta)} = \frac{\sin(\alpha) \cdot \cos(\beta) + \cos(\alpha) \cdot \sin(\beta)}{\cos(\alpha) \cdot \cos(\beta) - \sin(\alpha) \cdot \sin(\beta)} = \frac{\frac{\sin(\alpha) \cdot \cos(\beta)}{\cos(\alpha) \cdot \cos(\beta)} + \frac{\cos(\alpha) \cdot \sin(\beta)}{\cos(\alpha) \cdot \cos(\beta)}}{\frac{\cos(\alpha) \cdot \cos(\beta)}{\cos(\alpha) \cdot \cos(\beta)} - \frac{\sin(\alpha) \cdot \sin(\beta)}{\cos(\alpha) \cdot \cos(\beta)}} =$$

$$= \frac{\tan(\alpha) + \tan(\beta)}{1 - \tan(\alpha) \cdot \tan(\beta)}$$

12 Wie nachfolgend gezeigt ergibt sich daraus die Beziehung:

$$\arctan(\frac{1}{2}) + \arctan(\frac{1}{3}) = \frac{\pi}{4}$$

13 Beweis:
Für $\alpha_1 = \arctan(\frac{1}{2})$ und $\alpha_2 = \arctan(\frac{1}{3})$ folgt
$\tan(\alpha_1 + \alpha_2) = \frac{\tan(\alpha_1) + \tan(\alpha_2)}{1 - \tan(\alpha_1) \cdot \tan(\alpha_2)} = \frac{\frac{1}{2} + \frac{1}{3}}{1 - \frac{1}{2} \cdot \frac{1}{3}} = 1$
Weil zum einen aus $0 < \alpha_1 < \frac{\pi}{2}$ und $0 < \alpha_2 < \frac{\pi}{2}$ folgt, dass $0 < \alpha_1 + \alpha_2 < \pi$ ist
und zum andern $\alpha = \frac{\pi}{4}$ die einzige Lösung von $\tan(\alpha) = 1$ im Intervall $0 < \alpha < \pi$ ist,
folgt $\alpha_1 + \alpha_2 = \frac{\pi}{4}$ qed.

14 Folgerung $\pi = 4 \cdot (\arctan(\frac{1}{2}) + \arctan(\frac{1}{3}))$
Mit
$\arctan(x) = \sum_{k=0}^{\infty}(-1)^k \frac{x^{2k+1}}{2k+1} = \sum_{k=0}^{\infty}(\frac{x^{4k+1}}{4k+1} - \frac{x^{4k+3}}{4k+3})$ für $|x| < 1$

3 Dezimalzahlen

15 Den Typ Dezimalzahl definiere ich als (σ ist als Konstante definiert und bezeichnet die Anzahl Hinterkommastellen):

Type **Dezimalzahl**
  ganz As Long
  st(1 To sigma) As Integer
End Type

16 Ist $\sigma = 1000$, so wird dann die Zahl x=342.1374 abgelegt als:
x as Dezimalzahl, x.ganz=342, x.st(1)=1, x.st(2)=3, x.st(3)=7, x.st(4)=4, x.st(5)=0 usw. bis x.st(1000)=0

17 Wie beim schriftlichen Rechnen definiere ich die Addition und Subtraktion zweier Dezimalzahlen, sowie die Multiplikation einer Dezimalzahl mit einer ganzen Zahl und die Division einer Dezimalzahl durch eine ganze Zahl. Siehe Grundoperationen im nächsten Abschnitt.

18 Mit diesen vier Grundoperationen lassen sich die Berechnung aus Kapital 2 ausführen.

4 Programmtechnische Umsetzung in Excel Visual Basic

19 Deklarationen

```
Option Explicit

Const sigma = 1010

Type Dezimalzahl
  ganz As Long
  st(1 To sigma) As Integer
End Type
```

20 Hauptprogramm

```
Function CalcPi(ByVal stellen As Integer) As String
CalcPi = DezToStr(Multipliziere(Addiere(ArcTanRez(2), ArcTanRez(3)), 4), stellen)
End Function
```

21

```
Function ArcTanRez(ByVal n As Integer) As Dezimalzahl
Dim potenz As Dezimalzahl, AnzSummanden As Integer, k As Integer
AnzSummanden = n
potenz = Dividiere(IntToDez(1), n)
ArcTanRez = IntToDez(0)
For k = 0 To AnzSummanden
  ArcTanRez = Addiere(ArcTanRez, Dividiere(potenz, 4 * k + 1))
  potenz = Dividiere(potenz, n)
  potenz = Dividiere(potenz, n)
  ArcTanRez = Subtrahiere(ArcTanRez, Dividiere(potenz, 4 * k + 3))
  potenz = Dividiere(potenz, n)
  potenz = Dividiere(potenz, n)
Next k
End Function
```

22 Umwandlungen

```
Function IntToDez(ByVal x As Long) As Dezimalzahl
  IntToDez.ganz = x
End Function

Function DezToStr(a As Dezimalzahl, Optional ByVal stellen = sigma) As String
Dim k As Integer
If stellen > sigma Then
  stellen = sigma
End If
DezToStr = a.ganz & "."
For k = 1 To stellen
  If k Mod 5 = 1 Then
    DezToStr = DezToStr & " "
  End If
  DezToStr = DezToStr & a.st(k)
Next k
End Function
```

[23] Grundoperationen

```
Function Addiere(a As Dezimalzahl, b As Dezimalzahl) As Dezimalzahl
Dim k As Integer, uebertrag As Integer
uebertrag = 0
For k = sigma To 1 Step -1
  uebertrag = a.st(k) + b.st(k) + uebertrag
  Addiere.st(k) = uebertrag Mod 10
  uebertrag = (uebertrag - Addiere.st(k)) / 10
Next k
Addiere.ganz = a.ganz + b.ganz + uebertrag
End Function
```

```
Function Subtrahiere(a As Dezimalzahl, b As Dezimalzahl) As Dezimalzahl
Dim k As Integer, uebertrag As Integer
uebertrag = 0
For k = sigma To 1 Step -1
  If a.st(k) < b.st(k) + uebertrag Then
    Subtrahiere.st(k) = a.st(k) + 10 - (b.st(k) + uebertrag)
    uebertrag = 1
  Else
    Subtrahiere.st(k) = a.st(k) - (b.st(k) + uebertrag)
    uebertrag = 0
  End If
Next k
If a.ganz < b.ganz + uebertrag Then
  MsgBox ("Minuend kleiner Subtrahend ⇒ Abbruch")
  End
Else
  Subtrahiere.ganz = a.ganz - (b.ganz + uebertrag)
End If
End Function
```

```
Function Dividiere(a As Dezimalzahl, n As Integer) As Dezimalzahl
Dim k As Integer, rest As Integer, uebertrag As Integer
rest = a.ganz Mod n
Dividiere.ganz = (a.ganz - rest) / n
For k = 1 To sigma
  uebertrag = 10 * rest + a.st(k)
  rest = uebertrag Mod n
  Dividiere.st(k) = (uebertrag - rest) / n
Next k
End Function
```

```
Function Multipliziere(a As Dezimalzahl, n As Integer) As Dezimalzahl
Dim k As Integer, rest As Integer, uebertrag As Integer
uebertrag = 0
For k = sigma To 1 Step -1
  uebertrag = a.st(k) * n + uebertrag
  Multipliziere.st(k) = uebertrag Mod 10
  uebertrag = (uebertrag - Multipliziere.st(k)) / 10
Next k
Multipliziere.ganz = a.ganz * n + uebertrag
End Function
```

5 Fehlerabschätzung

24 Bezeichne z eine reelle Zahl, z' die auf σ Stellen abgerundete Zahl. Definiere $\delta := 10^{-\sigma}$
Es gilt $0 \leq z - z' < \delta$.
In Verallgemeinerung dessen gelte für die Näherung z' von z sogar nur $|z - z'| < \delta$

25 Grundrechenarten:
Addition:
$|(z_1 + z_2) - (z_1' + z_2')| = |z_1 + z_2 - z_1' - z_2'| \leq |z_1 - z_1'| + |z_2 - z_2'| < 2\delta$
Subtraktion:
$|(z_1 - z_2) - (z_1' - z_2')| = |z_1 - z_2 - z_1' + z_2'| \leq |z_1 - z_1'| + |z_2' - z_2| < 2\delta$
Was uns hier interessiert ist die Abweichung bei einer n-Gliedrigen Summe bzw. Differenz:
$|\sum\limits_{k=1}^{n} z_k - \sum\limits_{k=1}^{n} (z_k)'| < n \cdot \delta$
Multiplikation:
$|z_1 \cdot n - z_1' \cdot n| = |z_1 - z_1'| \cdot n < \delta \cdot n$
Division:
$|z_1/n - (z_1'/n)'| \leq |z_1/n - z_1'/n| + |z_1'/n - (z_1'/n)'| = |z_1 - z_1'|/n + |z_1'/n - (z_1'/n)'| < \delta/n + \delta = \frac{n+1}{n} \cdot \delta$
Was uns hier interessiert ist die Mehrfachdivision (durch ganzzahliges $n_i \geq 2$) mit fortlaufendem Abrunden:
Behauptung: $|\ldots((z/n_1)/n_2)\ldots/n_k - (\ldots((z'/n_1)'/n_2)'\ldots/n_k)'| < 2\delta$
Beweis durch Induktion:
Verankerung: Im Fall $k = 0$ ergibt sich $|z - z'| < \delta < 2\delta$ nach Voraussetzung.
Schritt: Die Behauptung gelte für k.

$$
\begin{aligned}
&|(\ldots((z/n_1)/n_2)\ldots/n_k)/n_{k+1} - ((\ldots((z'/n_1)'/n_2)'\ldots/n_k)'/n_{k+1})'| \leq \\
&\quad \leq |(\ldots((z/n_1)/n_2)\ldots/n_k)/n_{k+1} - (\ldots((z'/n_1)'/n_2)'\ldots/n_k)'/n_{k+1}| + \\
&\quad + |(\ldots((z/n_1)'/n_2)'\ldots/n_k)'/n_{k+1} - ((\ldots((z'/n_1)'/n_2)'\ldots/n_k)'/n_{k+1})'| < \frac{2\delta}{n_{k+1}} + \delta \leq 2\delta \quad \text{qed.}
\end{aligned}
$$

26 Es bezeichne

$$\text{ArcTanRez}(n) := \sum_{k=0}^{\infty} \left(\frac{1}{(4k+1) \cdot n^{4k+1}} - \frac{1}{(4k+3) \cdot n^{4k+3}} \right)$$

und

$$\text{ArcTanRez}_\sigma(n) := \sum_{k=0}^{\sigma} \left(\frac{1}{(4k+1) \cdot n^{4k+1}} - \frac{1}{(4k+3) \cdot n^{4k+3}} \right)$$

und
$\text{ArcTanRez}'_\sigma(n)$ den Wert, bei dem jeder der $2\sigma + 2$ Summanden von $\text{ArcTanRez}_\sigma(n)$ als Mehrfachdivision gemäss Punkt 25 entstanden ist.

27 Gemäss Punkt 25 ist $|\text{ArcTanRez}'_\sigma(n) - \text{ArcTanRez}_\sigma(n)| < (2\sigma + 2) \cdot 2 \cdot 10^{-\sigma}$

28 Für $n \geq 2$ gilt die Abschätzung:

$$
\begin{aligned}
&0 \leq \text{ArcTanRez}(n) - \text{ArcTanRez}_\sigma(n) = \\
&\quad = \sum_{k=\sigma+1}^{\infty} \left(\frac{1}{(4k+1) \cdot n^{4k+1}} - \frac{1}{(4k+3) \cdot n^{4k+3}} \right) \leq \sum_{k=\sigma+1}^{\infty} \frac{1}{(4k+1) \cdot n^{4k+1}} \leq \\
&\quad \leq \sum_{k=\sigma+1}^{\infty} \frac{1}{n^{4k+1}} \leq \sum_{k=\sigma+1}^{\infty} \frac{1}{2^{4k+1}} = \frac{1}{2^{4\sigma+5}} \cdot \frac{1}{1 - \frac{1}{2^4}} \leq \frac{1}{2^{4\sigma}} = \frac{1}{16^\sigma} \leq \frac{1}{10^\sigma} = 10^{-\sigma}
\end{aligned}
$$

29 Jetzt ergibt sich ($n \geq 2$):

$$
\begin{aligned}
&|\text{ArcTanRez}(n) - \text{ArcTanRez}'_\sigma(n)| = \\
&\quad = |\text{ArcTanRez}(n) - \text{ArcTanRez}_\sigma(n) + \text{ArcTanRez}_\sigma(n) - \text{ArcTanRez}'_\sigma(n)| \leq \\
&\quad \leq |\text{ArcTanRez}(n) - \text{ArcTanRez}_\sigma(n)| + |\text{ArcTanRez}_\sigma(n) - \text{ArcTanRez}'_\sigma(n)| \leq \\
&\quad \leq 10^{-\sigma} + (2\sigma + 2) \cdot 2 \cdot 10^{-\sigma} = (4\sigma + 5) \cdot 10^{-\sigma}
\end{aligned}
$$

30 Zusammengesetzt ergibt sich ($n \geq 2$):

$$|\pi - 4 \cdot (\text{ArcTanRez}'_\sigma(2) + \text{ArcTanRez}'_\sigma(3))| =$$
$$= |4 \cdot (\text{ArcTanRez}(2) + \text{ArcTanRez}(3)) - 4(\text{ArcTanRez}'_\sigma(2) + \text{ArcTanRez}'_\sigma(3))| \leq$$
$$\leq 4 \cdot |\text{ArcTanRez}(2) - \text{ArcTanRez}'_\sigma(2))| + 4 \cdot |\text{ArcTanRez}(3)) - \text{ArcTanRez}'_\sigma(3)| \leq$$
$$\leq 4 \cdot (4\sigma + 5) \cdot 10^{-\sigma} + 4 \cdot (4\sigma + 5) \cdot 10^{-\sigma} = (32\sigma + 40) \cdot 10^{-\sigma} = 32360 \cdot 10^{-1010} \leq 4 \cdot 10^{-1006}$$

für $\sigma := 1010$.

6 Resultat

31 Das Programm liefert mit Aufruf: ?CalcPi(1010)

3.
14159 26535 89793 23846 26433 83279 50288 41971 69399 37510 58209 74944 59230 78164 06286 20899 86280 34825 34211 70679
82148 08651 32823 06647 09384 46095 50582 23172 53594 08128 48111 74502 84102 70193 85211 05559 64462 29489 54930 38196
44288 10975 66593 34461 28475 64823 37867 83165 27120 19091 45648 56692 34603 48610 45432 66482 13393 60726 02491 41273
72458 70066 06315 58817 48815 20920 96282 92540 91715 36436 78925 90360 01133 05305 48820 46652 13841 46951 94151 16094
33057 27036 57595 91953 09218 61173 81932 61179 31051 18548 07446 23799 62749 56735 18857 52724 89122 79381 83011 94912
98336 73362 44065 66430 86021 39494 63952 24737 19070 21798 60943 70277 05392 17176 29317 67523 84674 81846 76694 05132
00056 81271 45263 56082 77857 71342 75778 96091 73637 17872 14684 40901 22495 34301 46549 58537 10507 92279 68925 89235
42019 95611 21290 21960 86403 44181 59813 62977 47713 09960 51870 72113 49999 99837 29780 49951 05973 17328 16096 31859
50244 59455 34690 83026 42522 30825 33446 85035 26193 11881 71010 00313 78387 52886 58753 32083 81420 61717 76691 47303
59825 34904 28755 46873 11595 62863 88235 37875 93751 95778 18577 80532 17122 68066 13001 92787 66111 95909 21642 01989
38095 25788

32 Die unsicheren Stellen (1005 bis 1010) sind unterstrichen!

33 Der Output wurde der bessern Lesbarkeit halber in 20 5er-Blöcke pro Zeile unterteilt.

Literaturverzeichnis

Gellert, W. / Küstner, H. / Hellwich, M. / Kästner, H. / Reichardt H. (1977): Kleine Enzyklopädie Mathematik. 10., völlig überarbeitete Auflage. Leipzig: VEB Verlag Enzyklopädie

Baloui, Said (1998): Excel 97 Kompendium. Überarbeitete Neuauflage. München: Markt & Technik Buch- und Software-Verlag Gmbh